

M. SZYSZKOWICZ (Wrocław)

**A PROCEDURE REALIZING A SECOND ORDER
ONE-STEP METHOD FOR SOLVING A SYSTEM
OF ORDINARY DIFFERENTIAL EQUATIONS**

1. Procedure declaration. The procedure *diffsystbobkov2* solves the initial value problem of the form

$$(1) \quad y'_k = f_k(x, y_1(x), y_2(x), \dots, y_n(x)),$$

$$(2) \quad y_k(x_0) = y_{0k} \quad (k = 1, 2, \dots, n)$$

at the points x_1, x_2, \dots

Data:

x — the value of x_0 in (2);

$x1$ — the value of the argument for which we solve the problem;

eps — the relative error (the given tolerance);

eta — the number which is used instead of zero when the obtained solution is zero or near to zero;

$hmin$ — the least admissible absolute value of the step length;

n — the number of differential equations in (1) and (2);

$y[1:n]$ — the values of the right-hand sides of (2).

Results:

x — the value of $x1$;

$y[1:n]$ — the values of the approximate solution $y_k(x)$ ($k = 1, 2, \dots, n$).

Additional parameters:

$steph$ — label outside of the body of the procedure *diffsystbobkov2* to which a jump is made if the absolute value of the step length is smaller than $hmin$; after the jump, x is equal to the value \tilde{x} ($\tilde{x} < x1$) for which the approximate solution has a relative error equal to the given eps , and $y[1:n]$ contains the value of this approximate solution;

f — identifier of the procedure which computes the values of the right-hand sides of (1) and puts them in $d[1:n]$; this procedure has the following heading: **procedure** $f(x, n, y, d)$; **value** x, n ; **real** x ; **integer** n ; **array** y, d .

```

procedure diffsysbobkov2(x,x1,eps,eta,hmin,n,y,steph,f);
  value x1,eps,eta,hmin,n;
  real x,x1,eps,eta,hmin;
  integer n;
  array y;
  label steph;
  procedure f;
  begin
    real h,hh,w,w1,w2,w3,w4;
    integer i;
    Boolean last;
    array d,df,dff,y1,y2[1:n];
    eps:=.16666666666666/eps;
    h:=x1-x;
    last:=true;
    f(x,n,y,d);
  conth:
    hh:=h*.5;
    w:=h*.375;
    for i:=1 step 1 until n do
      begin
        w1:=y[i];
        w2:=d[i];
        y1[i]:=w1+hh*w2;
        y2[i]:=w1+w*w2
      end i;
    f(x+hh,n,y1,df);
    f(x+w,n,y2,dff);
    hh:=h*.75;
    for i:=1 step 1 until n do

```

```

begin
  w1:=y[i];
  y1[i]:=w1+h×df[i];
  y2[i]:=w1+hh×dff[i]
  end i;
f(x+h,n,y1,df);
f(x+hh,n,y2,dff);
hh:=h×.5;
w2:=h×.3333333333333333;
w:=.0;
for i:=1 step 1 until n do
  begin
    w1:=d[i];
    w3:=w2×(w1+2.0×dff[i]);
    w1:=w3-hh×(w1+df[i]);
    w3:=y2[i]:=y[i]+w3+w1×.3333333333333333;
    w1:=abs(w1);
    w3:=abs(w3);
    if w3<eta
      then w3:=eta;
    w3:=w1/w3;
    if w3>w
      then w:=w3
    end i;
w:=if w=.0 then eta else 1.25×(eps×w)†.3333333333333333;
hh:=h/w;
if w>1.25
  then
    begin
      if abs(hh)<hmin

```

```

    then go to steph;
    last:=false
end w>1.25
else
begin
    x:=x+h;
    for i:=1 step 1 until n do
        y[i]:=y2[i];
    if last
        then go to endp;
    f(x,n,y,d);
    w:=x1-x;
    if (w-hh)*h<0
        then
        begin
            hh:=w;
            last:=true
        end(w-hh)*h<0
    end w<1.25;
    h:=hh;
    go to conth;
endp:
end diffesystbobkov2

```

2. Method used. In the construction of the procedure *diffesystbobkov2* we use two one-step methods of second order. The methods are of the form

$$(3) \quad \eta_{n+1} = \eta_n + \frac{h}{2} (f_n + f_{n+1})$$

with the additional formulas

$$\eta_{n+1/2} = \eta_n + \frac{h}{2} f_n, \quad \eta_{n+1} = \eta_n + hf_{n+1/2},$$

and

$$(4) \quad \eta_{n+1} = \eta_n + \frac{h}{3} (f_n + 2f_{n+3/4})$$

with the additional formulas

$$\eta_{n+3/8} = \eta_n + \frac{3}{8} hf_n, \quad \eta_{n+3/4} = \eta_n + \frac{3}{4} hf_{n+3/8},$$

where η_{n+a} is an approximate solution obtained at the point $x_n + ah$, and

$$f_{n+ah} = f(x_n + ah, \eta_{n+a}).$$

Using the results due to Bobkov [2], p. 38-42, we may treat the solution obtained by (4) as the solution calculated by method (3) twice with step $h/2$. In this way, we may use the solutions obtained by methods (3) and (4) to vary the step of integration.

In paper [3] the method (3) is used once with step h and twice with step $h/2$ for the control of the step size.

This step size control mechanism is described in [1]. The numerical results for the procedure *diffsystbobkov2* are better than those of the procedure *diffsystheun* (see [3]) which realizes the method (3).

3. Certification. The procedure *diffsystbobkov2* has been verified on the Odra 1204 computer for many examples of the initial value problem. Some of them are presented here.

Examples.

$$(A) \quad y_1' = 1/y_2, \quad y_1(0) = 1, \quad y_2' = -1/y_1, \quad y_2(0) = 1$$

with the exact solution $y_1 = e^x, y_2 = e^{-x}$.

$$(B) \quad y_1' = 10 \cos 10x, \quad y_1(0) = 0,$$

with the exact solution $y_1 = \sin 10x$.

$$(C) \quad \begin{cases} y_1' = 10 \operatorname{sgn}(\sin 20x) y_2, & y_1(0) = 0, \\ y_2' = -10 \operatorname{sgn}(\sin 20x) y_1, & y_2(0) = 1, \end{cases}$$

with the exact solution $y_1 = |\sin 10x|, y_2 = |\cos 10x|$.

In the sequel we present the obtained relative error and the number of evaluations of the function f ($[f]$).

The results were obtained for $eps = eta$ and $hmin = 10^{-15}$ at the points .5, 1.0, 1.5, and 10.0. Here we give only the results for $x = 1.5$ and $x = 10.0$.

The results obtained for the problem (A)

x	$eps = 10^{-3}$ [f]	$eps = 10^{-6}$ [f]	$eps = 10^{-9}$ [f]
1.5	1.5_{10}^{-5} 14 2.4_{10}^{-4}	1.2_{10}^{-7} 74 3.6_{10}^{-7}	4.4_{10}^{-10} 689 3.2_{10}^{-11}
10.0	-2.8_{10}^{-2} 128 3.8_{10}^{-2}	-3.2_{10}^{-5} 1173 4.1_{10}^{-5}	-3.2_{10}^{-8} 11613 4.1_{10}^{-8}

The results obtained for the problem (B)

x	$eps = 10^{-3}$ [f]	$eps = 10^{-6}$ [f]	$eps = 10^{-9}$ [f]
1.5	-7.4_{10}^{-4} 117	5.1_{10}^{-7} 728	4.1_{10}^{-9} 6942
10.0	-5.0_{10}^{-3} 2181	-3.7_{10}^{-6} 14217	9.5_{10}^{-8} 134643

The results obtained for the problem (C)

x	$eps = 10^{-3}$ [f]
1.5	-2.9_{10}^{-3} 941 -2.9_{10}^{-3}
10.0	-5.0_{10}^{-2} 15558 -5.0_{10}^{-2}

References

- [1] J. S. Chomicz, A. Olejniczak and M. Szyszkowicz, *A method for finding the step size of integration of a system of ordinary differential equations*, Zastos. Mat. 17 (1983), p. 645-654.
- [2] V. I. Крылов, V. V. Бобков and P. I. Монастырнуй (В. И. Крылов, В. В. Бобков и П. И. Монастырний), *Вычислительные методы*, т. 2, Москва 1977.
- [3] M. Szyszkowicz, *Two algorithms for solving the initial value problem by using Runge-Kutta-Heun's method*, Raport Nr N-53, Instytut Informatyki Uniwersytetu Wrocławskiego, Wrocław 1978.

INSTITUTE OF COMPUTER SCIENCE
UNIVERSITY OF WROCLAW
51-151 WROCLAW

Received on 16. 2. 1981